

CHAPTER 6. TERMINAL INPUT AND OUTPUT

The source code discussed in this chapter is in file KERNEL86.BLK, screens 41 to 49.

Forth is an interpretive language which intimately interacts with the user through a CRT terminal. Terminal input and output control is a very important part of the Forth system, allowing the user to enter commands and data into the computer and display the results or messages on the CRT. Many Forth implementations have input/output commands coded in the host machine codes which access the terminal interface directly. These Forth systems are often stand-alone system which do not need support from a traditional operating system. F83 was design to run under the popular CP/M or MS-DOS system, so that it can be transported between different host computers. The terminal input/output commands in F83 thus utilize the CP/M or DOS BIOS routines to receive information from the keyboard and send information to the CRT display.

6.1. THE BDOS I/O CALLS TO THE OPERATING SYSTEM

The fundamental interface between Forth terminal I/O commands and the CP/M or DOS is the Forth word BDOS:

CODE BDOS(entry function --- return-value)	Load function code into C register
	and entry parameter into D register. Call the BIOS. Return
	result are then pushed on the data stack.
CX POP	Load function code into C register.
DX POP	Load entry parameter into D register.
33 INT	Call BIOS by a software interrupt. This is the MS-DOS
	interrupt vector. For CP/M, it is 224 INT.
AH AH SUB	Clear the high byte in the AX register.
1PUSH	Return with the result on stack.
END-CODE	

BDOS is not only used for terminal I/O, it can also be used for most of the disk I/O calls, making Forth I/O commands very neat and simple.

: (KEY?) `(--- f)	Return a true flag if the user presses a key. Otherwise,
	return a false flag.
0 11 BDOS	Function 11 is the direct console I/O call. Entry 0 specifies
	a console status command. If no character is ready, 0 is
	returned. If a character is entered, FFH is returned.
0<>	Reversed the BDOS flag.
;	
: (KEY) (--- char)	Wait until a key is pressed, and then return the ASCII code.
BEGIN	Enter the wait loop.
PAUSE	Release the CPU to other tasks so that the multitasking
	scheme can work smoothly.

(KEY?) UNTIL 0 8 BDOS ;	Is a key pressed? Yes, then exit the loop. Otherwise, wait another round. Entry parameter 0 specifies a console input function and returns an ASCII code on the stack.
: (CONSOLE) (char ---) PAUSE 6 BDOS DROP 1 #OUT +! ;	Send the character on stack to the terminal for display. Let other task have a run on the CPU. Call BDOS to send out the character. BDOS always returns a number on the stack. It has to be dropped. Increment user variable #OUT, keeping track of the output character count.
: (PRINT) (char ---) PAUSE 5 BDOS DROP 1 #OUT +! ;	Send a character to the printer. Always pause before an I/O operation because I/O operations are generally slow. CPU can then be freed to serve other tasks or other users. Function 5 is the BDOS call for output to listing device. Clear the stack. Increment #OUT.
: (EMIT) (char ---) PRINTING @ IF DUP (PRINT) -1 #OUT +! (CONSOLE) ;	Send the character to both the terminal and the printer. Is the printing flag set? Yes. Sent character to the printer. Print character. Back up #OUT so it will not be incremented twice. THEN Output to the terminal.

(KEY?), (KEY), and (EMIT) are the actual operators vectored to by KEY?, KEY, and EMIT. EMIT can vector to (PRINT) or (EMIT) like the regular CP/M system to activate the printer with the console.

6.2. TERMINAL OUTPUT COMMANDS

The following output words are all simple derivatives of EMIT and they do not need extensive comments:

: CRLF (---) 13 EMIT 10 EMIT #OUT OFF 1 #LINE +! ;	Send a carriage return and a line feed to the console. Carriage return. Line feed. Clear the output character count. Increment the line count.
--	--

<pre> : TYPE (addr len ---) 0 ?DO DUP C@ EMIT 1+ LOOP DROP ; </pre>	<pre> Display a string on the console. Repeat len times, but skip if len is zero. Send one character. Increment character address. Clear stack. </pre>
<pre> : SPACE (---) BL EMIT ; </pre>	<pre> Send a space to console. </pre>
<pre> : SPACES (n ---) 0 MAX 0 ?DO SPACE LOOP ; </pre>	<pre> Send n spaces to console. Eliminate negative counts. Repeat n times. </pre>
<pre> : BACKSPACES (n ---) 0 ?DO BS EMIT LOOP ; </pre>	<pre> Send n backspaces to console. </pre>

6.3. INTERPRETING CONTROL CHARACTERS

Forth is capable of using most of the ASCII characters for word names. Only a few ASCII codes are reserved for system functions. Many fig-Forth system reserve the NUL (ASCII 0), CR (ASCII 13), and SP (ASCII 32) as delimiters for Forth words. The DEL (ASCII 127) is used to nullify the previously entered character, which is important in correcting typing errors. Other non-printable characters or control character can be used freely to name definitions. Because it is difficult to document the non-printable characters, embedding them in names is discouraged unless you want a very secured environment.

F83, on the other hand, provides a mechanism for you to implement special functions for control characters. EXPECT checks to see if an input character is a control character. When a function is defined for a particular control character, the function will be executed immediately when that character is entered on the keyboard. A jump table is maintained for all the 32 control characters. A few of them are used for special purposes which are defined as follows:

<pre> : BS-IN (n char --- n-1) DROP DUP IF 1- BS ELSE BELL THEN EMIT ; </pre>	<pre> Back up the input character buffer by dropping the character off the stack and decrementing n by 1. If n is zero, sound the bell instead. Used for ctrl-H. Discard the character. Is n=0? Yes. Decrement n and backspace. n=0. Sound the bell. Send either BS or BELL to console. </pre>
--	--

<pre> : (DEL-IN) (n char --- n-1) DROP DUP IF 1- BS SPACE BS ELSE BELL THEN EMIT ; </pre>	<p>Backup the input and erase the previous character. If n=0, sound the bell. Used for DEL (127).</p> <p>Backspace.</p> <p>Send a space and backspace again. Erase the previous character.</p>
<pre> : BACK-UP (n char --- 0) DROP DUP BACKSPACES DUP SPACES BACKSPACES 0 ; </pre>	<p>Erase the current line and set the character count to zero. Used for ctrl-U and ctrl-X.</p> <p>Discard the character on the stack.</p> <p>Backup to the beginning of the current line.</p> <p>Erase all the characters on this line.</p> <p>Backup</p> <p>Clear character count.</p>
<pre> : RES-IN (char ---) FORTH TRUE ABORT" Reset" ; </pre>	<p>Reset the Forth system to a clean start again. (ctrl-C)</p> <p>Set default vocabulary.</p> <p>Force system abort.</p> <p>Abort with a message.</p>
<pre> : P-IN (char ---) DROP PRINTING @ NOT PRINTING ! ; </pre>	<p>Toggle the printer on or off. (ctrl-P)</p> <p>Get the flag in PRINTING.</p> <p>Complement it to turn the printer on or off.</p> <p>Store it back.</p>
<pre> : CR-IN (m addr n char --- m addr m) DROP SPAN ! OVER BL EMIT ; </pre>	<p>Finish input and remember the number of characters in SPAN. (ctrl-M or CR)</p> <p>Store n in SPAN.</p> <p>Duplicate m.</p> <p>Send out a space.</p>
<pre> : (CHAR) (addr n char --- addr n+1) 3DUP EMIT + ! 1+ ; </pre>	<p>Process a normal character by appending it to the input buffer.</p> <p>Send character to console.</p> <p>Addr+n, the memory address for the current character.</p> <p>Store the character into the input buffer at addr+n.</p> <p>Increment n by 1 for the next character.</p>
<pre> DEFER CHAR DEFER DEL-IN </pre>	<p>CHAR will be vectored to (CHAR).</p> <p>DEL-IN will be vectored to (DEL-IN).</p>

VARIABLE CC	CC will be used to point to the current control character table.
CREATE CC-FORTH	The control character table which can handle each control character as a special case. It is actually an execution array which is indexed into by EXPECT to do the right thing when it receives a control character.
]	Enter compilation mode to compile 32 execution address for the 32 control characters.
CHAR CHAR CHAR CHAR CHAR CHAR CHAR CHAR	
BS-IN CHAR CHAR CHAR CHAR CR-IN CHAR CHAR	
P-IN CHAR CHAR CHAR CHAR BACK-UP CHAR CHAR	
BACK-UP CHAR RES-IN CHAR CHAR CHAR CHAR CHAR	
[Reenter the execution mode.

6.4. MORE SOPHISTICATED INPUT COMMANDS

KEY is the most elementary word to accept keyboard input. It simply gets a character and puts its ASCII code up on the data stack: not a very intelligent word. Once we have the control character table, we can build a very intelligent input definition which can respond to many control characters to do a wide range of different things in response to our keyboard strokes. This definition is EXPECT:

: EXPECT	(addr len ---)	Get a string from the terminal and place it in the buffer at addr specified. Perform a limited amount of line editing. Save the number of characters input in the variable SPAN. Process control characters as specified by the control character table pointed to by CC.
DUP SPAN !		Save len in SPAN.
SWAP 0		Stack is now: len addr 0 ---
BEGIN		Start the input loop.
2 PICK		Copy len to top of stack.
OVER -		(len addr count #left ---)
WHILE		If all characters have been received, exit the loop. If #left is not 0, continue on.
KEY		Get one more character.
DUP BL <		Is it less than 32, i.e., a control character?
IF		Yes. A control character.
DUP 2*		Offset into the CC table.
CC @ +		The table entry address.
PERFORM		Execute the CC table entry.
ELSE		Not a control character.
DUP 127 =		Is it a DEL?
IF DEL-IN		Yes. Do delete the prior character.
ELSE CHAR		No. A regular character. Append it to the input buffer.
THEN		
THEN		
REPEAT		End of string input loop.
2DROP DROP		Clear the stack.

<pre> ; : TIB (--- addr) 'TIB @ ; </pre>	<p>Get the address of the terminal input buffer. TIB is vectored through 'TIB.</p>
<pre> : QUERY (---) TIB 80 EXPECT SPAN @ #TIB ! BLK OFF >IN OFF ; </pre>	<p>Get an input stream of text from the terminal and store it in the terminal input buffer. Prepare the system to interpret this input text.</p> <p>Receive upto 80 characters into the terminal input buffer.</p> <p>Get the actual length of the input stream, which may be less than 80.</p> <p>Store it in #TIB so that the text interpreter will know when the text is exhausted.</p> <p>Clear BLK so that the text interpreter will use the terminal input buffer for text input.</p> <p>Clear the character pointer to start from the beginning of the terminal input buffer.</p>

QUERY is the Forth input word at the highest level. It waits on the user to type a line of text on the keyboard. The line is terminated either by receiving 80 characters from the keyboard or by receiving a carriage return key. The line of text is stored in the terminal input buffer. All the pertinent parameters are set so that the text interpreter can take over and interpret or execute the commands given in the input line.

6.5. STRING COMMANDS

Screens 41 to 43 are a set of commands to operate on strings in memory. A string in Forth is a sequence of ASCII characters preceded by a byte count. A string may have zero to 255 characters. It is generally identified by the address of the count byte. However, most string commands require the address of the first character in the string as argument, not the address of the count byte. String commands use the following generalized syntax:

<source addr> <dest addr> <length> <string command>

Destination address is optional in cases of single string operations.

Most of the string commands are standard Forth-83 words and their definitions are simple and straightforward. I will only list here their functions and stack parameters:

TABLE 6.1. STRING COMMANDS

COUNT (addr --- addr+1 len)	Convert the string address to address-length representation.
LENGTH (addr --- addr+2 len)	Return address-length for long strings whose character count is 16-bits.
FILL (addr len char ---)	Initialize a string to char.
ERASE (addr len ---)	Initialize a string to NUL.
BLANK (addr len ---)	Initialize a string to blanks.
MOVE (sour dest len ---)	Move a string without overlapping.
UPC (char --- char')	Convert a character to upper case.
UPPER (addr len ---)	Convert s string to upper case.
-TRAILING (addr len --- addr len')	Delete trailing blanks from a string by changing its length.
COMP (sour dest len --- n)	Compare source string with destination string. Return -1 if source<destination. Return 1 if source>destination. Return 0 if strings are the same.
CAPS-COMP (sour dest len --- n)	Compare two strings regardless of character cases.
COMPARE (sour dest len --- n)	Compare two strings. If CAPS is true, convert to upper case before comparing.

Figure 6.1 Representation of strings